
Linear Approximation and Newton's Method  

> restart;  

Linear Approximation  

We said in class that in order to simplify computations we can use a linear approximation of a function 

 , i.e., instead of working with  we use its linear approximation  

 .  

If we do this then the change in  -values , i.e.,  , is represented by the 

differential  , where  is proportional to  , the change in  .  

In particular,  is approximated by  

 .  

Obviously, this approximation introduces an error , i.e., we can't expect  to give us the exact value 

of  , and neither can we expect  (unless  of course).  

In order to illustrate this idea we take the function  and ask how well does the 

linear approximation of  at  represent  throughout the interval [  ].  

Therefore, we first define   

> f:=x->sqrt(x)-sin(x);  

  

Next we compute the linear approximation  of  at  .  

> x0:=2; 
L:=x->f(x0)+D(f)(x0)*(x-x0): 
'L(x)' = L(x);  

Here is a plot of both  and  on the interval [  ].  

> plot({f(x),L(x)}, x=0..2*Pi, color=[green,red]);  

  

As to be expected, we can see that the approximation is good only near the point  . 

This example is continued in Ex.1 below.  

Newton's Method  

From class we know that the Newton iteration formula is given by  

 .  

Remember that the purpose of Newton's method is to compute a zero of the (possibly nonlinear) 

function  based on an initial guess  , i.e.,  

Newton's method is used to solve the equation  for  .  

This can be programmed very nicely in Maple since the required derivative of  can be computed 

symbolically.  

Here is a function along with its roots given to us by Maple's built-in command fsolve .  

Note that fsolve uses Newton's method as one of its approaches to solving an equation numerically.  

> f:=x->x^3-3*x^2+1; 
Digits:=15: 
fsolve(f(x)=0, x);  

  

  

And here is an implementation of the Newton iteration.  

Note that besides some print statements a check has been added to catch when  since the 

method will obviously fail in this case.  

> Newton := proc (f, x0, N) 
### The parameters are 
### f function for which we are trying to find the zero 
### x0 initial guess 
### N maximum number of iterations allowed 
### 
local k, df; 
k := 0: 
x.k := x0; 
printf(" k x[k]\n\n"); 
printf("%2d %15.12f \n", k, x.k); 
df := D(f); 
to N-1 while abs(evalf(f(x.k))) > 10^(1-Digits) do 
if df(x.k) <>0 then 
x.(k+1) := evalf( x.k - f(x.k)/df(x.k) ); 
else 
printf("\n Abort: D(f)(x[%d]) = 0\n", k); 
break; 
fi; 
printf("%2d %15.12f \n", k+1, x.(k+1)); 
k := k+1: 
od; 
RETURN(x.k); 
end:  

Here we use Newton with a starting value of  and see that this results in an approximate 

root .65...  

> Newton(f, 1.6, 10):  

 k        x[k] 

 0   1.600000000000   

 1    .254166666667   

 2    .872119301312   

 3    .662542256243   

 4    .652740866146   

 5    .652703645213   

 6    .652703644666   

To get one of the other roots we should use a different starting value:  

> Newton(f, -1, 10): 
Newton(f, 2.2, 10):  

 k        x[k] 

 0  -1.000000000000   

 1   -.666666666667   

 2   -.548611111111   

 3   -.532390161865   

 4   -.532088989397   

 5   -.532088886238   

 6   -.532088886238   

 k        x[k] 

 0   2.200000000000   

 1   4.375757575758   

 2   3.499052876838   

 3   3.047211126488   

 4   2.896960878278   

 5   2.879610054371   

 6   2.879385279075   

 7   2.879385241572   

Strange this can happen though!  

You would probably not expect the following outcome:  

> Newton(f, 1.8, 10):  

 k        x[k] 

 0   1.800000000000   

 1   -.874074074074   

 2   -.614029754316   

 3   -.538726323286   

 4   -.532138380314   

 5   -.532088889023   

 6   -.532088886238   

And sometimes the method fails.  

> Newton(f, 2, 10):  

 k        x[k] 

 0   2.000000000000   

    Abort: D(f)(x[0]) = 0 

This is clear if you look at the iteration formula since 2 is a zero of the derivative of  :  

> solve(D(f)(x)=0);  

  

Graphical Illustration of Newton's Method  

The following modification of the basic Newton method produces a movie that graphically illustrates 
the method's progress.  

The additional parameters  and  are used to define an interval over which the function will be 

plotted.  

> Newton_graph := proc(f, x0, N, a, b) 
local x, df, i, ymin, ymax, pf, tangent; 
i := 0: 
x := x0; 
df := D(f); 

ymin := minimize(f(z),{z}, {z=a..b}): 
ymax := maximize(f(z),{z}, {z=a..b}): 
pf := plot(f, a-(b-a)/20..b+(b-a)/20, view=[a..b, ymin*1.05..ymax*1.05]): 
l.i := plottools[line]([x,0], [x,evalf(f(x))], color=blue): 
tangent := unapply( df(x)*(z-x) + f(x), z); 
t.i := plot(tangent, a..b, color=green): 
p.i := plots[display](pf, l.i, t.i): 
to N-1 while abs(evalf(f(x))) > 10^(1-Digits) do 
i := i+1: 
x := evalf( x - f(x)/df(x) ); 
l.i := plottools[line]([x,0], [x,evalf(f(x))], color=blue): 
tangent := unapply( df(x)*(z-x) + f(x), z); 
t.i := plot(tangent, a..b, color=green): 
p.i := plots[display](pf, seq({l.j, t.j}, j=0..i)): 
od; 
plots[display](seq(p.j, j=0..i), insequence=true); 
end:  

Here is a possible use of this procedure.  

We use the same function as above.  

(Click on the plot, and then use the VCR controls to activate the animation)  

> Newton_graph(f, 1.8, 10, -1, 2);  

  

A Poor Man's Fractal  

As we saw above, unexpected outcomes of Newton's method are possible.  

If we treat the function  as a complex-valued function the situation is even more amazing.  

Here we use the same function as above and run Newton with initial guesses taken from a grid of points 
in the complex plane.  

Depending on which of the three roots the method converges to we color the starting point with a 
different color, i.e.,  

if you take any green point as a starting value Newton's answer is close to the root 2.88,  

for a dark blue point the answer is .65,  

and for a light blue point we have found the root -.53.  

Red points indicate that the method failed (or did not come close enough to one of the roots within the 
allowed number of steps).  

> fractal := proc(a, b, c, d, n, N) 
### The parameters are 
### a, b left and right boundary for x-dimension 
### c, d lower and upper boundary for y-dimension 
### n resolution of image 
### N maximum number of iterations for Newton method 
### 
local f, df, pointlist, colorlist, dotcolor, x, y, z, i, j; 
Digits:=7; 
f := z -> (z^3-3*z^2+1); 
df := D(f); 
pointlist := []: 
colorlist := []: 
for i from 0 to n do # loop over x-direction of image 
for j from 0 to n do # loop over y-direction of image 
x := a+i*(b-a)/n; 
y := c+j*(d-c)/n; 
#### NEWTON #### 
z := x + I*y; 
to N while abs(f(z)) > 10^(1-Digits) do 
if df(z) <> 0 then 
z := z - f(z)/df(z); 
else 
break; 
fi; 
od: 
#### END NEWTON #### 
if abs(z +.53) < .02 then 
dotcolor := cyan; # convergence to root -.53 
elif abs(z - .65) < .02 then 
dotcolor := blue; # convergence to root .65 
elif abs(z - 2.88) < .02 then 
dotcolor := green; # convergence to root 2.88 
else  
dotcolor := red; # no apparent convergence  

fi; 
pointlist := [op(pointlist),[[x, y]]]; 
colorlist := [op(colorlist), dotcolor]; 
od: 
od: 
plot(pointlist, color=colorlist, axes=frame, style=point, symbol=box); 
end:  

Note: you might have to reduce  (=30) a little if your computer cannot handle the problem.  

> fractal(1.6, 2.4, -.3, .3, 30, 10);  

  

> fractal(1.64, 1.68, -.015, .015, 30, 10);  

  

Assignment 7  

Ex.1:  

Consider the linear approximation example begun above.  

a) Compute the absolute error  . (Make sure that you redefine  ).  

b) Plot this function over the interval [  ].  

c) Find the maximum of  .  

d) How far can we move away from  so that the error is no larger than  ?  

Mathematically this question means: how large can we pick  such that  implies 

 , where  . 

You may answer this question graphically.  

Ex.2:  

When the length  of a clock pendulum is held constant by controlling its temperature, the pendulum's 

period  depends on the acceleration of gravity  . The period will therefore vary slightly as the 

clock is moved from place to place on the earth's surface, depending on the change in  . By keeping 

track of  , we can estimate the variation of  from the equation  that relates  

,  , and  .  

a) With  held constant and  as the independent variable, calculate  and use it to answer the 

next two questions.  

b) If g increases, will T increase or decrease? Will a pendulum clock speed up or slow down? Explain.  

c) A clock with a  pendulum is moved from a location where  to a new location. 

This increases the period by  . Find  and estimate the value of  at the new location. 

Ex.3:  

You are asked by a friend who knows you are good at calculus to verify a survey report he recently 

obtained. In this report the length of a certain property line was listed as  . You put together 

some crude surveying instruments and head out to the location. The instruments you have consist of two 

poles. One of them is  tall, and the other  . On top of the lower pole you place an optical 

instrument that can measure angles to within 5% of the true angle. If the angle you read off is  , 

what are you going to tell your friend? (Assume that the ground is level, which - in Illinois - is a pretty 
safe assumption).  

Ex.4:  

a) Produce another plot of the fractal image above by zooming in at the leftmost blue point.  

b) Based on your image, select starting points in the interval [  ] for each one of the four 

possible outcomes (i.e., convergence to -.53, .65, 2.88, and no apparent convergence) of the (real) 
Newton method and call the Newton procedure with these starting values to verify your guesses.  

c) Use the graphical version of the Newton method to show illustrate how the iteration proceeds in each 
of your four examples.  

Ex.5:  

The moon of Planet Gzyx has an elliptical orbit with eccentricity 0.5, and its period of revolution around 
the planet is 100 days.  

If the moon is at the position (  ) when  (see the figure below), then the central angle after 

 days is given by Kepler's equation  

 .  

Use Newton's method to solve for  when  days. Take  (radians).  

> with(plottools): 
a := 1: b := sqrt(3)/2: x0 := 0: y0:=0: 
elli := ellipse([x0,y0], a, b, color=red): 
circ := ellipse([x0,y0], a, a, color=green): 
planet := plot([[1/2,0]], style=point, symbol=circle, color=blue, thickness=3): 
moon := plot([[1/2,3/4]], style=point, symbol=circle, color=cyan, thickness=3): 
lines := plot([[[0,0], [1/2,sqrt(3)/2]], [[1/2,0], [1/2,sqrt(3)/2]] ], color=black, linestyle=3): 
moontext := plots[textplot]([1/2+.1,3/4+.1,`Moon`],align={ABOVE,RIGHT}, color=cyan): 

planettext := plots[textplot]([1/2+.05,0.05,`Gzyx`],align={ABOVE,RIGHT}, color=blue): 
angletext := plots[textplot]([.1,0.05,`q`], font=[SYMBOL,10], align={ABOVE,RIGHT}, 
color=black): 
plots[display]({elli, circ, planet, moon, lines, moontext, planettext, angletext}, scaling=constrained, 
axes=normal);  

